NAG Fortran Library Routine Document

F07BUF (CGBCON/ZGBCON)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07BUF (CGBCON/ZGBCON) estimates the condition number of a complex band matrix A, where A has been factorized by F07BRF (CGBTRF/ZGBTRF).

2 Specification

```
 SUBROUTINE F07BUF(NORM, N, KL, KU, AB, LDAB, IPIV, ANORM, RCOND, WORK,
1                  RWORK, INFO)
 ENTRY      cgbcon (NORM, N, KL, KU, AB, LDAB, IPIV, ANORM, RCOND, WORK,
1                  RWORK, INFO)
 INTEGER           N, KL, KU, LDAB, IPIV(*), INFO
 real              ANORM, RCOND, RWORK(*)
 complex           AB(LDAB,*), WORK(*)
 CHARACTER*1       NORM
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine estimates the condition number of a complex band matrix A, in either the 1-norm or the infinity-norm:

$$\kappa_1(A) = \|A\|_1 \|A^{-1}\|_1 \quad \text{or} \quad \kappa_\infty(A) = \|A\|_\infty \|A^{-1}\|_\infty.$$

Note that $\kappa_\infty(A) = \kappa_1(A^H)$.

Because the condition number is infinite if A is singular, the routine actually returns an estimate of the **reciprocal** of the condition number.

The routine should be preceded by a call to F06UBF to compute $\|A\|_1$ or $\|A\|_\infty$, and a call to F07BRF (CGBTRF/ZGBTRF) to compute the LU factorization of A. The routine then uses Higham's implementation of Hager's method (see Higham (1988)) to estimate $\|A^{-1}\|_1$ or $\|A^{-1}\|_\infty$.

4 References

Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with applications to condition estimation *ACM Trans. Math. Software* **14** 381–396

5 Parameters

1: NORM – CHARACTER*1 *Input*

On entry: indicates whether $\kappa_1(A)$ or $\kappa_\infty(A)$ is estimated as follows:

if NORM = '1' or 'O', $\kappa_1(A)$ is estimated;

if NORM = 'I', $\kappa_\infty(A)$ is estimated.

Constraint: NORM = '1', 'O' or 'I'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $N \geq 0$.

3: KL – INTEGER *Input*

On entry: k_l, the number of sub-diagonals within the band of A.

Constraint: $KL \geq 0$.

4: KU – INTEGER *Input*

On entry: k_u, the number of super-diagonals within the band of A.

Constraint: $KU \geq 0$.

5: AB(LDAB,*) – ***complex*** array *Input*

Note: the second dimension of the array AB must be at least $\max(1, N)$.

On entry: the LU factorization of A, as returned by F07BRF (CGBTRF/ZGBTRF).

6: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07BUF (CGBCON/ZGBCON) is called.

Constraint: $LDAB \geq 2 \times KL + KU + 1$.

7: IPIV($*$) – INTEGER array *Input*

Note: the dimension of the array IPIV must be at least $\max(1, N)$.

On entry: the pivot indices, as returned by F07BRF (CGBTRF/ZGBTRF).

8: ANORM – ***real*** *Input*

On entry: if NORM = '1' or 'O', the 1-norm of the **original** matrix A; if NORM = 'I', the infinity-norm of the **original** matrix A. ANORM may be computed by calling F06UBF with the same value for the parameter NORM. ANORM must be computed either **before** calling F07BRF (CGBTRF/ZGBTRF) or else from a **copy** of the original matrix A.

Constraint: $ANORM \geq 0.0$.

9: RCOND – ***real*** *Output*

On exit: an estimate of the reciprocal of the condition number of A. RCOND is set to zero if exact singularity is detected or the estimate underflows. If RCOND is less than ***machine precision***, A is singular to working precision.

10: WORK($*$) – ***complex*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, 2 * N)$.

11: RWORK($*$) – ***real*** array *Workspace*

Note: the dimension of the array RWORK must be at least $\max(1, N)$.

12: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed estimate RCOND is never less than the true value ρ, and in practice is nearly always less than 10ρ, although examples can be constructed where RCOND is much larger.

8 Further Comments

A call to this routine involves solving a number of systems of linear equations of the form $Ax = b$ or $A^H x = b$; the number is usually 5 and never more than 11. Each solution involves approximately $8n(2k_l + k_u)$ real floating-point operations (assuming $n \gg k_l$ and $n \gg k_u$) but takes considerably longer than a call to F07BSF (CGBTRS/ZGBTRS) with 1 right-hand side, because extra care is taken to avoid overflow when A is approximately singular.

The real analogue of this routine is F07BGF (SGBCON/DGBCON).

9 Example

To estimate the condition number in the 1-norm of the matrix A, where

$$A = \begin{pmatrix} -1.65+2.26i & -2.05-0.85i & 0.97-2.84i & 0.00+0.00i \\ 0.00+6.30i & -1.48-1.75i & -3.99+4.01i & 0.59-0.48i \\ 0.00+0.00i & -0.77+2.83i & -1.06+1.94i & 3.33-1.04i \\ 0.00+0.00i & 0.00+0.00i & 4.48-1.09i & -0.46-1.72i \end{pmatrix}.$$

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F07BUF Example Program Text
*     Mark 15 Release. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, KLMAX, KUMAX, LDAB
      PARAMETER        (NMAX=8,KLMAX=8,KUMAX=8,LDAB=2*KLMAX+KUMAX+1)
      CHARACTER        NORM
      PARAMETER        (NORM='1')
*     .. Local Scalars ..
      real             ANORM, RCOND
      INTEGER          I, INFO, J, K, KL, KU, N
*     .. Local Arrays ..
      complex          AB(LDAB,NMAX), WORK(2*NMAX)
      real             RWORK(NMAX)
      INTEGER          IPIV(NMAX)
*     .. External Functions ..
      real             F06UBF, X02AJF
      EXTERNAL         F06UBF, X02AJF
*     .. External Subroutines ..
      EXTERNAL         cgbcon, cgbtrf
*     .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07BUF Example Program Results'
*     Skip heading in data file
```

```
      READ (NIN,*)
      READ (NIN,*) N, KL, KU
      IF (N.LE.NMAX .AND. KL.LE.KLMAX .AND. KU.LE.KUMAX) THEN
*
*        Read A from data file
*
         K = KL + KU + 1
         READ (NIN,*) ((AB(K+I-J,J),J=MAX(I-KL,1),MIN(I+KU,N)),I=1,N)
*
*        Compute norm of A
*
         ANORM = F06UBF(NORM,N,KL,KU,AB(KL+1,1),LDAB,RWORK)
*
*        Factorize A
*
         CALL cgbtrf(N,N,KL,KU,AB,LDAB,IPIV,INFO)
*
         WRITE (NOUT,*)
         IF (INFO.EQ.0) THEN
*
*           Estimate condition number
*
            CALL cgbcon(NORM,N,KL,KU,AB,LDAB,IPIV,ANORM,RCOND,WORK,
     +                  RWORK,INFO)
*
            IF (RCOND.GE.X02AJF()) THEN
               WRITE (NOUT,99999) 'Estimate of condition number =',
     +           1.0e0/RCOND
            ELSE
               WRITE (NOUT,*) 'A is singular to working precision'
            END IF
         ELSE
            WRITE (NOUT,*) 'The factor U is singular'
         END IF
      END IF
      STOP
*
99999 FORMAT (1X,A,1P,e10.2)
      END
```

9.2 Program Data

```
F07BUF Example Program Data
  4  1  2                                                 :Values of N, KL and KU
 (-1.65, 2.26) (-2.05,-0.85) ( 0.97,-2.84)
 ( 0.00, 6.30) (-1.48,-1.75) (-3.99, 4.01) ( 0.59,-0.48)
               (-0.77, 2.83) (-1.06, 1.94) ( 3.33,-1.04)
                             ( 4.48,-1.09) (-0.46,-1.72)  :End of matrix A
```

9.3 Program Results

```
 F07BUF Example Program Results

 Estimate of condition number =  1.04E+02
```